Seismic Analysis Code (SAC) – Bullard Tutorial

by Keith Priestley

A. Overview:

The package of computer routines making up the *Seismic Analysis Code* [**SAC**] was developed by Joe Tull and Bill Tapley at the Lawrence Livermore National Laboratory, University of California, in the early 1980's. It is a command-driven program implemented on many operating systems and has become a standard data manipulating tool for seismology. The **SAC** binary data format is one of the principle data formats used in seismological research. For a detailed description of the **SAC** data format see http://www.iris.edu/files/sac-manual/manual/file_format.html. With **SAC** one can interactively read seismogram data and perform many standard data processing operations, such as:

- filtering time series
- picking times and amplitudes
- removing or correcting seismograph response
- determining particle motion and rotating seismograms
- de-glitching seismograms
- computing the Fourier transform of a time series
- computing the envelope of a time series
- determining spectral ratios
- performing linear regressions
- performing frequency-wavenumber analysis
- creating a variety of plots of the data and the resulting analysis

There are also utility programs that can be used to view plots or to convert **SAC** graphics files into other formats such as PostScript for incorporating into *LaTex* document files. One can also interact directly with the **LINUX** operating system while using **SAC**, calling system functions or running secondary programs. Scripts of **SAC** commands (**macros**) can be executed directly within **SAC** or combined with **bash** scripts for automating processing. **SAC** binary data files can be written and read from **FORTRAN** or **C** programs using input/output routines contained in the **SAC** library. The IRIS website has a *SAC Software Manual* [http://www.iris.edu/files/sac-manual/] and *SAC Tutorial Guide* [http://www.iris.edu/software/sac/manual/tutorial.html]. This tutorial is meant to be an introduction to **SAC**; you should use these websites as your primary reference for **SAC**. In this tutorial keith@jupiter:~ $ is the computer prompt – your prompt will be set up differently. Anything in green is the computer response; what you type on the keyboard, enter through a bash script or enter through a **SAC** macro is shown in black. The character # signifies to bash that anything following is a comment.

B. Setting up SAC to run from your account:

SAC is installed at jupiter:/usr/local/bin/sac. After login into jupiter type "sac" in the command line. If you get something like

```
keith@jupiter:~ $sac
SEISMIC ANALYSIS CODE [08/20/2013 (Version 101.6)]
Copyright 1995 Regents of the University of California

   SAC>
```

then **SAC** is working; if **SAC** does not start up it is likely you need to add the **SAC** executable to your PATH and set up two environmental variables. Type the following:

```
keith@jupiter:~ $ echo $PATH
```

and you should see "/usr/local/bin" somewhere in your path. Typing the following:

```
keith@jupiter:~ $ echo $SACDIR
```

should give

```
keith@jupiter:~ $ /usr/local/sac
```

and typing

```
keith@jupiter:~ $ echo $SACAUX
```

should give

```
keith@jupiter:~ $  /usr/local/sac/aux
```

If your do not see these, add the following lines in your $HOME/.bashrc files to make **SAC** work properly.

```
# added for Seismic Analysis Code (SAC)
export PATH=/usr/local/bin:$PATH
export SACDIR=/usr/local/sac
export SACAUX=$SACDIR/aux
```

C. Getting started:

To get a feel for **SAC** try the following. The lines beginning with an ∗ are comments. In most cases you do not need to type the full **SAC** command.

```
keith@jupiter:~ $sac
SEISMIC ANALYSIS CODE [08/20/2013 (Version 101.6)]
Copyright 1995 Regents of the University of California

SAC>∗turn decimation off using QDP
SAC>qdp off
SAC>∗ generated an impulse time series 1024 points long using FUNCGEN
SAC>fg n 1024
SAC>∗ turn off the file id when plotting using FILEID
SAC>fileid off
SAC>∗ plot lines in the color red using COLOR (note spelling of COLOR)
SAC>color on inc on l red
SAC>∗ increase the line width of the plot using WIDTH
SAC>width 2
SAC>∗ make the plot annotation a little nicer using XDIV
SAC>xdiv nice power off
SAC>∗ begin the plot frame with BEGINFRAME
SAC>beginframe
SAC>∗ set the plotting window with XVPORT and YVPORT
SAC>xvp .1 .3
SAC>yvp .3 .6
SAC>∗ plot the impulse time series with PLOT1
SAC>p1
SAC>xvp 0.4 0.6
SAC>∗ apply a 4 pole butterworth with a 0.02-0.20 Hz BANDPASS
SAC>bp butter co 0.02 0.2 n 4
SAC>∗ plot the filtered impulse
SAC>p1
SAC>∗ compute the fft of the filtered impulse using FFT
SAC>fft
SAC>xvp 0.7 0.9
SAC>∗ plot the amplitude spectrum with PLOTSP
SAC>psp am
SAC>∗ end the plotting frame with ENDFRAME
SAC>endframe
SAC>∗ close sac with QUIT
SAC>quit
```

You should get the following:



The left hand plot is the impulse computed with FUNCGEN, the centre plot is the impulse after applying a four pole butterworth BANDPASS filter with a passband from 0.02-0.20 Hz and the right hand plot is the amplitude spectrum of the bandpass filtered impulse computed with FFT and plotted with PLOTSP.

D. Seismogram example:

```
keith@jupiter:~ $sac
SEISMIC ANALYSIS CODE [08/20/2013 (Version 101.6)]
Copyright 1995 Regents of the University of California

SAC>*read a vertical component broadband seismogram
SAC>r 2007013042321/2007013042321_CU_GTOK.00.R.BHZ
```

The **SAC** header contains information about the seismogram: the sampling interval, start time, length, station location, event location, components, *etc.* You can examine the waveform header using the **SAC** command LISTHDR:

```
SAC>*examine the sac header content using LISTHDR
SAC>lh
FILE: 2007013042321/2007013042321_CU_GTOK.00.R.BHZ - 1
________________________________________
NPTS = 72001                      # number of data points
B = -1.907349e-06                 # begin time
E = 3.600000e+03                  # end time
IFTYPE = TIME SERIES FILE         # file type
LEVEN = TRUE                      # evenly sampled time series
DELTA = 5.000000e-02              # time increment
IDEP = UNKNOWN                    # physical unit of the data
DEPMIN = -9.895849e+05            # minimum amplitude
DEPMAX = 9.944349e+05             # maximum amplitude
DEPMEN = -4.400651e-01            # mean amplitude
OMARKER = 0                       # event origin marker
T1MARKER = 567.76 (P)             # predicted P arrival time
T3MARKER = 1027.6 (S)             # predicted S arrival time
KZDATE = JAN 13 (013), 2007       # reference date
KZTIME = 04:23:21.100             # reference time
IZTYPE = EVENT ORIGIN TIME        # type of reference time
KSTNM = GTOK                      # station name
CMPAZ = 0.000000e+00              # component azimuth relative to north
CMPINC = 0.000000e+00             # component "incidence angle" relative to the vertical
STLA = 2.736550e+01               # station latitude
STLO = 8.856910e+01               # station longitude
STEL = 0.000000e+00               # station elevation
EVLA = 4.624000e+01               # event latitude
EVLO = 1.545200e+02               # event longitude
EVDP = 1.000000e+01               # event depth
IEVTYP = EARTHQUAKE               # event type
DIST = 6.058422e+03               # source receiver distance in km
AZ = 2.739746e+02                 # azimuth
BAZ = 5.112683e+01                # back azimuth
GCARC = 5.449913e+01              # back azimuth
LOVROK = TRUE                     # TRUE if it is okay to overwrite this file on disk
USER0 = 6.546142e-02              # ray parameter
USER7 = 7.300000e+00              # magnitude
NVHDR = 6                         # Header version number. Current value is the integer 6.
NORID = 0                         # Origin ID (CSS 3.0)
NEVID = 0                         # Event ID (CSS 3.0)
NWFID = 7                         # Waveform ID (CSS 3.0)
LPSPOL = TRUE                     # TRUE if station components have a positive polarity (left-hand rule)
LCALDA = TRUE                     # DIST, AZ, BAZ, and GCARC calculated from STLA and STLO
KCMPNM = HHZ                      # Component name
KNETWK = CU                       # seismic network
```

The # and following are comments and will not appear. Other header values may be set in your data files. A detailed description of the **SAC** header structures is given at http://www.iris.edu/files/sac-manual/manual/file_format.html. You can change a header value using the **SAC** command CHNHDR:

```
SAC>*change a sac header value using CHNHDR
SAC>ch LOVROK FALSE
SAC>lh LOVROK

FILE: 2007013042321/2007013042321_CU_GTOK.00.R.BHZ - 1
________________________________________

LOVROK = FALSE
```

You can compare the broadband and bandpassed waveform:

```
SAC>qdp off
SAC>xdiv nice power off
SAC>fileid off
SAC>xdiv nice
SAC>*only plot the bottom axes using AXES
SAC>axes only bottom
SAC>*only plot the bottom ticks using TICKS
SAC>ticks only bottom
SAC>*window the plot between 550 and 2750 seconds using XLIM
SAC>xlim 500 2750
SAC>beginframe
SAC>xvp 0.05 0.45
SAC>yvp .3 .6
SAC>p1
SAC>xvp 0.55 0.95
SAC>*remover the trend and mean using RTREND
SAC>rtrend
SAC>*taper the time-series TAPER
SAC>taper
SAC>bp butter co 0.005 0.025 n 4
SAC>p1
SAC>endframe
```



The left plot is the broadband seismogram and the right plot is the bandpass filtered seismogram. The predicted arrival times from the global velocity model ak135 for the P- and S-wave are indicated. Now execute the following **SAC** input lines to compare the horizontal components as recorded (N and E) with then horizontal components rotated to radial and transverse:

```
SAC>beginframe
SAC>*read a horizontal component broadband seismogram
SAC>r 2007013042321/2007013042321_CU_GTOK.00.R.BHE 2007013042321/2007013042321_CU_GTOK.00.R.BHN
SAC>xvp 0.20 0.80
SAC>yvp 0.525 0.90
SAC>*make the y-axis scaling the same for all times series using YLIM
SAC>ylim all
SAC>p1
SAC>*rotate the seismograms to radial and transverse using ROTATE
SAC>rotate
SAC>yvp 0.10 0.475
SAC>p1
SAC>endframe
SAC>* write out the rotated seismograms using WRITE
SAC>write 2007013042321/2007013042321_CU_GTOK.00.R.BHR 2007013042321/2007013042321_CU_GTOK.00.R.BHT
```



The left two traces are the east (upper) and north (lower) components of motion; the right two traces are the radial (upper) and transverse (lower) showing the separation of the Rayleigh Wave on the radial component and the Love Wave on the transverse component. You can verify the Rayleigh Wave particle motion in two ways.

```
SAC>beginframe
SAC>*read the vertical and radial component broadband seismogram
SAC>r 2007013042321/2007013042321_CU_GTOK.00.R.BHZ 2007013042321/2007013042321_CU_GTOK.00.R.BHR
SAC>rtrend
SAC>taper
SAC>*set the window to plot the Rayleigh Wave
SAC>xlim 1800 2100
SAC>bp co .01 .033 n 3
SAC>xvp 0.1 0.3
SAC>color on inc on l black red
SAC>*plot the two waveforms over each other     SAC>p2
SAC>xvp 0.40 0.60
SAC>plotpm
SAC>xvp 0.7 0.9
SAC>r 2007013042321/2007013042321_CU_GTOK.00.R.BHR
SAC>hilbert
SAC>r more 2007013042321/2007013042321_CU_GTOK.00.R.BHZ
SAC>rtrend
SAC>taper
SAC>bp co .01 .033 n 3
SAC>color on inc on l red black
SAC>p2
SAC>endframe
```







The left-hand plot of the vertical (black) and radial (red) component of the Rayleigh Wave shows the $\pi/2$ phase shift between the vertical and radial components. The middle plot shows the retrograde elliptical motion of the Rayleigh Wave. The Rayleigh Wave particle motion is also demonstrated in the right-hand plot where the Hilbert transform of the radial component (red) is plotted over the vertical component (black) waveform.

Finally, it is often important to correct or remove the instruments response and this can be accomplished in **SAC**. The instrument response can be defined in a number of ways but the most common is a **SEED RESP** file (kept in jupiter:/raid2/data/RESP) or a **SAC POLEZERO** files (kept in jupiter:/raid2/data/SACPZ). You can examine the instrument response as follows:

```
SAC>fg n 1024
SAC>*convolve the impulse with a sac polezero file using TRANSFER
SAC>transfer from none to polezero subtype SACPZ/SAC_PZs_CU_GTOK_BHZ
SAC>fft
SAC>psp am
```



This plot shows the amplitude response of the seismometer at GTOK which is specified by the **SAC** polezero file SAC_PZs_CU_GTOK_BHZ. The response is flate from the Nyquest frequency to ~0.02 Hz and then drops off at low frequencies.

```
SAC>beginframe
SAC>r 2007013042321/2007013042321_CU_GTOK.00.R.BHZ
SAC>rtrend
SAC>taper
SAC>xvp 0.05 0.45
SAC>xlim 500 2750
SAC>yvp 0.30 0.60
SAC>p1
SAC>transfer from polezero subtype SACPZ/SAC_PZs_CU_GTOK_BHZ to none FREQlimits .005 .01 .05 .0667
SAC>xvp 0.55 0.95
SAC>p1
SAC>endframe
```



E. SAC macros:

You can write set of **SAC** commands to be executed together in a file – a **SAC** macro. This makes work a lot more efficient if you need to perform the same task many times. All of the **SAC** commands given in the last example can be entered into an ascii text file using a text editor, for example **simple.sm** below, and executed as a **SAC** macro. As a simple example, the following is a **SAC** macro to create the preceding plot:

```
keith@jupiter:~ $ more simple.sm
* echos sac commands to screen while executing
echo on
qdp off
xdiv nice power off
fileid off
xdiv nice
axes only bottom
ticks only bottom
r 2007013042321/2007013042321_CU_GTOK.00.R.BHZ
beginframe
rtrend
taper
xvp 0.05 0.45
xlim 500 2750
yvp .3 .6
p1
transfer from polezero subtype SACPZ/SAC_PZs_CU_GTOK_BHZ FREQlimits .005 .01 .05 .0667
xvp 0.55 0.95
p1
endframe
```

To produce the previous plot type:

```
keith@jupiter:~ $ sac simple.sm
```

Much more complicated processing tasks can be accomplished using the blackboard variables, keys and arithmetic features of **SAC** and by combining **SAC** macros with bash or csh scripts.

F. SAC blackboard variables:

G. SAC arithmetic;

H. Linking with FORTRAN and C routines: